cgemod

ANARRES

Tax Instruments & Revenues in ANARRES_t

1

cgemod

Outline

- Introduction
- Tax Instruments
  - Base (price)
  - Behaviour
- Calibration of Tax Rates
- Price equations with taxes
- Tax Revenue Variables

2

cgemod


cgemod
Tax Instruments
Global CGE, 2025
© cgemod
3

3


cgemod
Tax Instruments
Instrument
Name
Base Price
Behaviour
Import duties
$TM_{w,c,r}$
$PWM_{w,c,r}$
ad valorem
Export taxes
$TE_{c,w,r}$
$PWE_{c,w,r}$
ad valorem
Sales taxes (GST)
$TS_{c,r}$
$PQS_{c,r}$
ad valorem
Value added tax
$TV_{c,r}$
$PQD_{c,r}$
ad valorem
Production taxes
$TX_{a,r}$
$PX_{a,r}$
ad valorem
Factor use taxes
$TF_{f,a,r}$
$WFA_{f,a,r}$
ad valorem
HH Income tax
$TYH_{h,r}$
$YH_{h,r}$
ad valorem
Factor income tax
$TYF_{f,r}$
$YF_{f,r}$ *(1 - $deprec_{f,r}$)
ad valorem
Global CGE, 2025
© cgemod
4

4

cgemod

Calibration of Tax Rates

5

cgemod

Calibration – Commodity Taxes

```
tm0(w,c,r)$cmr(w,c,r) =
 SUM[ tmr$map_w_tmr(w,tmr), SAMG(tmr,c,r)] /
  {SAMG(w,c,r) +
     SUM[cp$ct(cp,r),IMPMARG(w,cp,c,r)]} ;
```

$$\frac{vtm_{w,c,r}}{(VWM_{w,c,r} + VMARG_{w,c,r})}$$

From: worksheet 'maps'

map_w_tmr(w,tmr)	
wafr	tmafr
woecd	tmoecd
wrow	tmrow
wglo	tmglo

6

cgemod

Calibration – Commodity Taxes

```
te0(c,w,r)$cer(c,w,r) =
  SUM[ ter$map_w_ter(w,ter), SAMG(ter,c,r)]
                              /SAMG(c,w,r) ;
```

$$\frac{vte_{c,w,r}}{VWE_{c,w,r}}$$

From: worksheet 'maps'

map_w_ter(w,ter)	
wafr	teafr
woecd	teoecd
wrow	terow
wglo	teglo

7

Calibration – Commodity Taxes (GST)

```
ts0(c,r)$[SUM(a,SAMG(c,a,r))
          + SUM(h,SAMG(c,h,r))
          + SAMG(c, "govt",r)
          + SAMG(c, "i_s",r)
          - SAMG("saltax",c,r)] =

SAMG("saltax",c,r)/[SUM(a,SAMG(c,a,r))
                    + SUM(h,SAMG(c,h,r))
                    + SAMG(c,"govt",r)
                    + SAMG(c,"i_s",r)
                      - SAMG("saltax",c,r)
                      - SAMG("vattax",c,r)] ;
```

$$\frac{vts_{c,r}}{VQS_{c,r}}$$

8

Calibration – Commodity Taxes (VAT)

```
tv0(c,r)$[SUM(h, SAMG(c,h,r))
          - SAMG("vattax",c,r)]
   = SAMG("vattax",c,r)/
          [SUM(h, SAMG(c,h,r))
            - SAMG("vattax",c,r)] ;
```

$$\frac{vtv_{c,r}}{VQCD_{c,r}}$$

9

Calibration – Activity Taxes

```
tx0(a,r)$SAMG("total",a,r)
   = SAMG("prodtax",a,r)/ SAMG("total",a,r) ;
```

$$\frac{vtx_{a,r}}{VQX_{a,r}}$$

```
tf0(f,a,r)$SAMG(f,a,r)
   = SUM[ tff$map_f_tff(f,tff),SAMG(tff,a,r)]
                              /SAMG(f,a,r) ;
```

$$\frac{vtf_{f,a,r}}{VYF_{f,a,r}}$$

From: worksheet 'maps'

map_f_tff(f,tff)	
fland	tfland
fUskil	tfUskil
fskil	tfskil
fcap	tfcap

10

Calibration – Household Income Taxes

```
tyh0(h,r)$SAMG("total",h,r)
     = SAMG("dirtax",h,r)/SAMG("total",h,r) ;
```

$$\frac{vtyh_{h,r}}{VYH_{h,r}}$$

NB1: the choice of *SAMG*("*total*",*h*,*r*) = *YH*(*h*,*r*) as the base will be important when calibrating the household savings rate

NB2: one household but indexed on *h*
– this avoids problems if users choose to change the default account name for households and eases extensibility
– allows extension to model with multiple households

11

Calibration – Factor Income Taxes

```
tyf0(f,r)$[SAMG("total",f,r) - SAMG("i_s",f,r)]
    = SAMG("dirtax",f,r)
       /[SAMG("total",f,r) - SAMG("i_s",f,r)] ;
```

$$\frac{vtyf_{f,r}}{VDYF_{f,r}}$$

NB1: the choice of *(SAMG("total",f,r) –SAMG("i_s",f,r))* as the base indicates factor income net of factor savings (depreciation)

12

Prices with Taxes - Commodities

```
PERDEF2(c,w,r)$cer(c,w,r)..
   PER(c,w,r) =E=
      PWE(c,w,r)*[1 - TE(c,w,r) ]*ER(r) ;
```

```
PMRDEF2(w,c,r)$cmr(w,c,r)..
   PMR(w,c,r) =E=
      PWM(w,c,r)*[1 + TM(w,c,r) ]*ER(r) ;
```

13

Prices with Taxes - Commodities

```
PQDDEF(c,r)$((cd(c,r) OR cm(c,r)) AND rgn(r))..
          PQD(c,r) =E=
                  PQS(c,r) * [1 + TS(c,r) ] ;
```

```
PQCDDEF(c,r)$(cd(c,r) OR cm(c,r) AND rgn(r))..
           PQCD(c,r) =E=
                   PQD(c,r) * [1 + TV(c,r) ] ;
```

14

Prices with Taxes - Activities

```
PVADEF(a,r)$rgn(r)..
  PX(a,r)*QX(a,r)*[1 - TX(a,r) ]
          =E= (PVA(a,r)*QVA(a,r)) +
               (PINT(a,r)*QINT(a,r)) ;
```

NB: *PVA*(*a*,*r*) is the amount per unit output available to pay factors used by activity *a* in region *r*

15

Prices with Taxes -- Activities

```
QVAFOC(f,a,r)$(deltava(f,a,r) AND rgn(r))..
   WF(f,r)*WFDIST(f,a,r)*[1 + TF(f,a,r)]

        =E=

  PVA(a,r)*QVA(a,r)*(SUM(fp$deltava(fp,a,r),
    deltava(fp,a,r)
       *(ADFD(f,a,r)*FD(fp,a,r))
         **(rhova(a,r)))  **(-1)
           *deltava(f,a,r)
              *ADFD(f,a,r)**(-rhova(a,r))
                 *FD(f,a,r)**(-rhova(a,r)-1) ;
```


16

Income After Taxes & Savings

```
HEXPEQ(h,r)$rgn(r)..
HEXP(h,r) =E=
      {YH(h,r)*[ 1 - TYH(h,r) ]}
       *(1 - SHH(h,r)) ;
```

NB: The different calibration wrt to where the tax is applied

```
YFDISTEQ(f,r)$rgn(r)..
YFDIST(f,r) =E=
      {YF(f,r) - [deprec(f,r)*YF(f,r)]}
         *[ 1 - TYF(f,r) ] ;
```


17

Tax Revenue Variables

18

Why Tax Revenue Variables?

- Unnecessary
  - the model is fine without them
- Reasons
  - Summary variables
  - Target variables
  - Can make the model easier to read
  - Habit?

19

Tax Revenue Equations

```
MTAXEQ(r)$rgn(r)..
    MTAX(r) =E= SUM((c,w), TM(w,c,r)
                  *PWM(w,c,r)*ER(r)*QMR(w,c,r)) ;

ETAXEQ(r)$rgn(r)..
    ETAX(r) =E= SUM((c,w), TE(c,w,r)
                  *PWE(c,w,r)*ER(r)*QER(c,w,r)) ;

STAXEQ(r)$rgn(r)..
    STAX(r) =E= SUM(c,TS(c,r)*PQS(c,r)
                * (QINTD(c,r) + SUM(h,QCD(c,h,r))
                  + QGD(c,r) + QINVD(c,r))) ;

VTAXEQ(r)$rgn(r)..
    VTAX(r) =E= SUM((c,h),TV(c,r)*PQD(c,r)*QCD(c,h,r)) ;
```


20

cgemod

Tax Revenue Equations

```
ITAXEQ(r)$rgn(r)..
    ITAX(r) =E=  SUM(a,TX(a,r)*PX(a,r)*QX(a,r)) ;

FTAXEQ(r)$rgn(r)..
     FTAX(r) =E= SUM((f,a),
            TF(f,a,r)*WF(f,r)*WFDIST(f,a,r)*FD(f,a,r)) ;


FYTAXEQ(r)$rgn(r)..
    FYTAX(r) =E= SUM(f,TYF(f,r)
                  *(YF(f,r) - (deprec(f,r)*YF(f,r)))) ;

HTAXEQ(r)$rgn(r)..
     HTAX(r) =E=  SUM(h,TYH(h,r)*YH(h,r)) ;
```


21

cgemod ANARRES

Tax Instruments & Revenues

The End

22

cgemod